<?xml version="1.0" encoding="UTF-8"?>
<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform"
  xmlns:parser="http://kadm.com.ua/XMLSchema/Parser" xmlns="http://kadm.com.ua/XMLSchema/Grammar"
  xmlns:t="http://kadm.com.ua/XMLSchema/Indented-text" xmlns:set="http://kadm.com.ua/XMLSchema/Set">
  <xsl:output method="xml" indent="yes"/>
  <xsl:strip-space elements="*"/>

  <xsl:template match="text()"/>

  <xsl:template match="@*"/>

  <xsl:template match="/parser:sets">
    <t:file>
      <t:l>array(</t:l>
      <t:block>
        <xsl:for-each select="*">
          <t:l>
            <xsl:value-of select="@parser:number"/>
            <xsl:text> =&gt; </xsl:text>
            <xsl:apply-templates select="."/>
            <xsl:text>,</xsl:text>
          </t:l>
        </xsl:for-each>
      </t:block>
      <t:l>)</t:l>
    </t:file>
  </xsl:template>

  <xsl:template match="set:union">
    <xsl:text>new UnionSet(</xsl:text>
    <xsl:for-each select="*">
      <xsl:if test="position() > 1">
        <xsl:text>, </xsl:text>
      </xsl:if>
      <xsl:apply-templates select="."/>
    </xsl:for-each>
    <xsl:text>)</xsl:text>
  </xsl:template>

  <xsl:template match="set:single">
    <xsl:text>new SingleSet(</xsl:text>
    <xsl:apply-templates/>
    <xsl:text>)</xsl:text>
  </xsl:template>

  <xsl:template match="set:range">
    <xsl:text>new RangeSet(</xsl:text>
    <xsl:for-each select="*">
      <xsl:if test="position() > 1">
        <xsl:text>, </xsl:text>
      </xsl:if>
      <xsl:apply-templates select="."/>
    </xsl:for-each>
    <xsl:text>)</xsl:text>
  </xsl:template>

  <xsl:template match="set:item[@set:type = 'char']">
    <xsl:text>str::getCode('</xsl:text>
    <xsl:value-of
      select="translate(translate(text(), '\', '\\'), &quot;'&quot;, &quot;\'&quot;)"/>
    <xsl:text>')</xsl:text>
  </xsl:template>

  <xsl:template match="set:item[@set:type = 'dec']">
    <xsl:value-of select="text()"/>
  </xsl:template>

  <xsl:template match="set:item[@set:type = 'hex']">
    <xsl:text>0x</xsl:text>
    <xsl:value-of select="text()"/>
  </xsl:template>
</xsl:stylesheet>
